

**blabel\_option** — Option for labeling bars

[Syntax](#)   
 [Description](#)   
 [Option](#)   
 [Remarks and examples](#)   
 [Also see](#)

## Syntax

```
graph { bar | hbar } ... , ... blabel(what [ , where_and_how ]) ...
```

<i>what</i>	Description
<code>none</code>	no label; the default
<code>bar</code>	label is bar height
<code>total</code>	label is cumulative bar height
<code>name</code>	label is name of <i>yvar</i>
<code>group</code>	label is first <code>over()</code> group

<i>where_and_how</i>	Description
<code>position(outside)</code>	place label just above the bar ( <code>bar</code> ) or just to its right ( <code>hbar</code> )
<code>position(inside)</code>	place label inside the bar at the top ( <code>bar</code> ) or at rightmost extent ( <code>hbar</code> )
<code>position(base)</code>	place label inside the bar at the bar's base
<code>position(center)</code>	place label inside the bar at the bar's center
<code>gap(<i>relativesize</i>)</code>	distance from position
<code>format(<i>%fmt</i>)</code>	format if <code>bar</code> or <code>total</code>
<code>textbox_options</code>	look of label

See [\[G-4\] \*relativesize\*](#) and [\[G-3\] \*textbox\\_options\*](#).

## Description

Option `blabel()` is for use with `graph bar` and `graph hbar`; see [\[G-2\] `graph bar`](#). It adds a label on top of or inside each bar.

## Option

`blabel(what, where_and_how)` specifies the label and where it is to be located relative to the bar. *where\_and\_how* is optional and is documented under [Suboptions](#) below. *what* specifies the contents of the label.

`blabel(bar)` specifies that the label be the height of the bar. In

```
. graph bar (mean) empcost, over(division) blabel(bar)
```

the labels would be the mean employee cost.

`blabel(total)` specifies that the label be the cumulative height of the bar. `blabel(total)` is for use with `graph bar`'s `stack` option. In

```
. graph bar (sum) cost1 cost2, stack over(group) blabel(total)
```

the labels would be the total height of the stacked bar—the sum of costs. Also, the `cost1` part of the stack bar would be labeled with its height.

`blabel(name)` specifies that the label be the name of the *yvar*. In

```
. graph bar (mean) y1 y2 y3 y4, blabel(name)
```

The labels would be “mean of y1”, “mean of y2”, . . . , “mean of y4”. Usually, you would also want to suppress the legend here and so would type

```
. graph bar (mean) y1 y2 y3 y4, blabel(name) legend(off)
```

`blabel(group)` specifies that the label be the name of the first `over()` group. In

```
. graph bar cost, over(division) over(year) blabel(group)
```

the labels would be the name of the divisions. Usually, you would also want to suppress the appearance of the division labels on the axis:

```
. graph bar cost, over(division, axis(off)) over(year) blabel(group)
```

## Suboptions

`position()` specifies where the label is to appear.

`position(outside)` is the default. The label appears just above the bar (`graph bar`) or just to its right (`graph hbar`).

`position(inside)` specifies that the label appear inside the bar, at the top (`graph bar`) or at its rightmost extent (`graph hbar`).

`position(base)` specifies that the label appear inside the bar, at the bar’s base; at the bottom of the bar (`graph bar`); or at the left of the bar (`graph hbar`).

`position(center)` specifies that the label appear inside the bar, at its center.

`gap(relativesize)` specifies a distance by which the label is to be offset from its location (`outside`, `inside`, `base`, or `center`). The default is usually `gap(1.7)`. The `gap()` may be positive or negative and you can specify, for instance, `gap(*1.2)` and `gap(*.8)` to increase or decrease the gap by 20%; see [G-4] *relativesize*.

`format(%fmt)` is for use with `blabel(bar)` and `blabel(total)`; it specifies the display format to be used to format the height value. See [D] *format*.

*textbox\_options* are any of the options allowed with a `textbox`. Important options include `size()`, which determines the size of the text; `box`, which draws a box around the text; and `color()`, which determines the color of the text. See [G-3] *textbox\_options*.

## Remarks and examples

[stata.com](http://www.stata.com)

`blabel()` serves two purposes: to increase the information content of the chart (`blabel(bar)` and `blabel(total)`) or to change how bars are labeled (`blabel(name)` and `blabel(group)`).

Remarks are presented under the following headings:

*Increasing the information content*  
*Changing how bars are labeled*

## Increasing the information content

Under the heading *Multiple bars (overlapping the bars)* in [G-2] [graph bar](#), the following graph was drawn:


```
. graph bar (mean) tempjuly tempjan, over(region)
    bargap(-30)
    legend(label(1 "July") label(2 "January"))
    ytitle("Degrees Fahrenheit")
    title("Average July and January temperatures")
    subtitle("by regions of the United States")
    note("Source: U.S. Census Bureau, U.S. Dept. of Commerce")
```

To the above, we now add

```
blabel(bar, position(inside) format(%9.1f))
```

which will add the average temperature to the bar, position the average inside the bar, at the top, and format its value by using %9.1f:

```
. graph bar (mean) tempjuly tempjan, over(region)
    bargap(-30)
    legend(label(1 "July") label(2 "January"))
    ytitle("Degrees Fahrenheit")
    title("Average July and January temperatures")
    subtitle("by regions of the United States")
    note("Source: U.S. Census Bureau, U.S. Dept. of Commerce")
    blabel(bar, position(inside) format(%9.1f) color(white)) ← new
```



We also specified the *textbox\_option* `color(white)` to change the color of the text; see [G-3] [textbox\\_options](#). Dark text on a dark bar would have blended too much.

## Changing how bars are labeled


Placing the labels on the bars works especially well with horizontal bar charts:

```
. use http://www.stata-press.com/data/r13/nls88, clear
(NLSW, 1988 extract)
```

```

. graph hbar (mean) wage,
  over(occ, axis(off) sort(1))
  blabel(group, position(base) color(bg))
  ytitle("")
  by(union,
    title("Average Hourly Wage, 1988, Women Aged 34-46")
    note("Source: 1988 data from NLS, U.S. Dept. of Labor,
      Bureau of Labor Statistics")
  )

```



What makes moving the labels from the axis to the bars work so well here is that it saves so much horizontal space.

In the above command, note the first two option lines:

```

  over(occ, axis(off) sort(1))
  blabel(group, position(base) color(bg))

```

`blabel(group)` puts the occupation labels on top of the bars, and suboption `position(base)` located the labels at the base of each bar. We specified `over(,axis(off))` to prevent the labels from appearing on the axis. Let us run through all the options:

```
over(occ, axis(off) sort(1))
```

Specified that the chart be done over occupation, that the occupation labels not be shown on the axis, and that the bars be sorted by the first (and only) *yvar*, namely, (mean) wage.

```
ytitle("")
```

Specified that the title on the numerical *y* axis (the horizontal axis in this horizontal case) be suppressed.

```
by(union, title(...) note(...))
```

Specified that the entire graph be repeated by values of variable `union`, and specified that the title and note be added to the overall graph. Had we specified the `title()` and `note()` options outside the `by()`, they would have been placed on each graph.

## Also see

[G-2] [graph bar](#) — Bar charts